
	u "data/panel_mafiaconsmig", clear
	
	drop if iddecade==1|iddecade==4 // 1951 | 1981

		
**-----------------------------------------------------

			* Appendix B
			*Descriptives and validation

**-----------------------------------------------------

*Table B.1: Descriptive Statistic

estpost summ newspc newslasso newsrforest newspc_c newsestseq news_winsor news_lag zempcp zfirmpc zssc zssc_wlo zsscf change20 change20s above10k zmigshare zmigshare_currentpop zssmig zssmig_rob  zssmig_wlo  zem zem_currentpop zss zssf zss_rob zss_wloc pop1871 ipop 
	esttab using "tables/tableB1.tex", cell((N mean(fmt(%9.3f)) sd(fmt(%9.3f)) min(fmt(%9.3f)) max(fmt(%9.3f)))) nonumber nomtitle replace label
	
*Table B.2: Cities with and without mafia-related news, 1951

	bys municipalities: egen anymafia=max(newspc)
	replace anymafia=1 if anymafia>0 & anymafia!=.
	
	balancetable anymafia z_ce_poppre z_ce_familysizepre z_ce_analphpre z_ce_genderdiffeducpre z_ce_emp_pcFpre z_ce_emp1pre z_ce_emp2pre z_emppc_pre z_ce_emp3pre z_ce_emp4pre z_ce_vecchiaiapre z_ce_elderlydeppre z_ce_youngdeppre z_ce_ratioMpre z_ce_servicespre z_ce_nopropertyhousepre z_ce_inhabperroompre z_ce_lackservicepre using "tables/tableB2.tex",  replace varlabels stddiff ctitles("No mafia news" "Mafia news" "Diff Mean" "Std Difference")
	
	
*Figure B.1: Number of migrants from the south moving to the center and north, 1955–2000
preserve
u "data/migrsouth_tot", clear
tw (line p1p2_ year), ytitle("Number of southern migrants") xtitle("") xline(1958)
graph export "figures/figureB1.pdf", as(pdf) replace
restore 
	
*Figure B.2: Mafia and construction employment over time, 1961–1981

preserve
	u "data/panel_mafiaconsmig", clear // keep iddecade==4
	drop if iddecade==1
	
	*share cities with mafia-related news in a decade
	gen newsd=(news>0 & news!=.)
	bys iddecade: gen x=sum(newsd)
	bys iddecade: egen tot_citymafia=max(x)
	replace tot_citymafia=(tot_citymafia/5963)*100
	
	*total news and cities with news by decade
	collapse (sum) news (firstnm) tot_citymafia, by(iddecade)
	
	twoway scatter news iddecade , mcolor(plb3) msymbol(circle) || connected news iddecade, legend(off) xtitle(" ") name(ev1, replace) ytitle("Number of mafia-related news") scheme(plotplainblind) lcolor(blue) lpattern(solid) xlab(2 "1961" 3 "1971" 4 "1981")
	
	twoway scatter tot_citymafia iddecade , mcolor(plb3) msymbol(circle) || connected tot_citymafia iddecade, legend(off) xtitle(" ") name(ev2, replace) ytitle("% cities with mafia-related news") scheme(plotplainblind) lcolor(blue) lpattern(solid) xlab(2 "1961" 3 "1971" 4 "1981")
	
		graph combine ev1 ev2, cols(2)
		graph export "figures/figureB2.pdf", replace as(pdf)
restore

*Figure B.3: Population shares of construction employees (1951) and migrants from the South (1955)
preserve
	keep if iddecade==2
	format empcp Mdsouth55 %5.3f

	spmap empcp using "data/city-coord.dta", id(id) fcolor(Blues) ocolor(none none none none none none)  ///
			clmethod(quantile) clnumber(6) ///
			name(con, replace) graphregion(color(white)) title("Employed in construction 1951") 
	spmap Mdsouth55 using "data/city-coord.dta", id(id) fcolor(Greens) ocolor(none none none none none none) ///
			clmethod(quantile) clnumber(6) ///
			name(mig, replace) graphregion(color(white)) title("Migrants from the South, 1955") 

			graph combine con mig, cols(2)
			graph export "figures/figureB3.pdf", as(pdf) replace
restore		


*Figure B.4: Relation between population and mafia-related news per capita
preserve
bys municipalities: egen any_newspc=max(newspc)
keep if iddecade==2
	
tw (scatter newspc lnpop)(lpoly newspc lnpop, ytitle(max value of Mafia related news per capita 1961-81) xtitle(Log population 1961) legend(off))
graph export "figures/figureB4.pdf", replace
restore 

*Section B.1: Validation comparing year 1969 for La Stampa and Corriere

u "data/validation_corriere_1969.dta", clear

* Table B.3: Balance between cities with mafia-news coverage by La Stampa and not Corriere in 1969 (Validation)

preserve
	keep if instampa_orcorriere==1
	gen corriere_stampa=1 if d_corriere==1
	replace corriere_stampa=0 if d_lastampa==1
	balancetable corriere_stampa z_* using "tables/tableB3.tex", replace varlabels stddiff
restore

*Figure B.5: News items mentioning mafia in 1969 in La Stampa and Corriere


spmap d_lastampa using "data/city-coord.dta", id(id) fcolor(red) osize(vvthin vvthin vvthin) ///
	point(data("data/validation_corriere_1969.dta") by(d_corriere) xcoord(x_cc) ycoord(y_cc) fcolor(blue%50) osize(*.5) size(*.5) legenda(on)) graphregion(color(white)) 

	graph export "figures/figureB5.pdf", as(pdf) replace

	
